	Formulare – Tips und Tricks

Lisa Slater Nicholls

Dieser Artikel enthält eine Sammlung von Techniken und Ratschlägen zum Formularmanagement. Wenn Sie sich diese Techniken aneignen, sammeln Sie kleine Wissenseinheiten an, die zusammengenommen mehr bringen als die Summe der Einzelinformationen.


Wann ist ein Formular kein Formular?

Obwohl ich geschrieben habe, daß wir das Formularmanagement behandeln, besteht doch für unseren Zweck kein echter Unterschied zwischen Formularobjekten und Fenstern, die mit der älteren DEFINE WINDOW Xbase-Syntax erstellt wurden. Sie haben sicher den Artikel „Einmal arbeiten – mit mehreren Objekten“ gelesen. In diesem Artikel sahen wir ein Browse, das sich genau wie ein Grid verhielt, als wir dem Browse-Kommando die NAME-Klausel hinzufügten, um eine Objektreferenz auf das Browse zu erhalten. Dementsprechend können Sie eine NAME-Klausel auch in der DEFINE WINDOW-Anweisung verwenden, um eine Objektreferenz auf das Fenster, das Sie erstellen, zu erhalten. Mit Hilfe dieser Objektreferenz können Sie in jedem DEFINEd WINDOW die Methoden des Formulars aufrufen (zum Beispiel Show()) oder die Eigenschaften des Formulars verändern (zum Beispiel Caption).

Abbildung 1

Auch wenn Sie die NAME-Klausel nicht verwenden, werden Sie feststellen, daß ein DEFINEd WINDOW viele Attribute mit einem Formular gemeinsam hat. Da die DEFINEd WINDOWs ein Teil der _SCREEN.Forms werden, können Sie eine Referenz benutzen, um mit dem DEFINEd WINDOW Informationen auszutauschen, genauso als wenn Sie ein Formular adressieren. Um sich selbst zu überzeugen, probieren Sie die folgenden Codezeilen aus, deren Ergebnis in Abbildung 1 zu sehen ist:

CLEAR ALL

* _SCREEN.FormCount enthält 0

DEFINE WINDOW x FROM 1,1 TO 20,20

* Beachten Sie, daß ich die NAME-
* Klausel nicht benutzt habe

_SCREEN.Forms(_SCREEN.FormCount).;
Caption = “Mein eigenes Fenster“

_SCREEN.Forms(_SCREEN.FormCount).;
Show()

Wie Sie später in M_FORM.PRG sehen werden, kann es signifikant sein, daß die SCREEN.Forms DEFINEd WINDOWs referenzieren. In diesem Artikel benutze

ich die Begriffe „Formular“ und „Fenster“ synonym. Damit ist aber nicht gemeint, daß Sie alle Fenster in Visual FoxPro als Formularobjekte betrachten können. VFP enthält in _SCREEN.Forms() kein „Systemfenster“, wie es mit MODIFY COMMAND oder MODIFY FILE erstellt wird. Dafür können Sie aber mit der Formularreferenz-Syntax Systemfenster direkt adressieren.

Wie auch immer, viele Systemfenster von VFP benutzen ähnliche Strategien für die Positionierung und Größenanpassung. Ein Beispiel mit einem einfachen WAIT WINDOW können Sie in M_CWAIT.PRG sehen.

Automatisch zentrierte Formulare

Wenn Sie schon einmal versucht haben, während der Entwicklungszeit die AutoCenter-Eigenschaft anzupassen, werden Sie wissen, wie diese Anpassung die weitere Arbeit am Layout des Formulars erschwert. Ein Formular „lebt“ bereits während der Entwicklungszeit – und wenn ein Formular im VFP-Hauptfenster zentriert ist, kann es bei kleineren Formularen vollständig außerhalb des Formulareditierfensters plaziert sein. 

Um dieses Problem zu lösen, werden Ihnen die meisten Menschen vorschlagen, AutoCenter im Eigenschaften-Fenster doch einfach auf .F. zu stellen. Wenn Sie dann ein Fenster zentrieren möchten, könnten Sie AutoCenter in der Init()-Methode Ihres Formulars oder Ihrer Formularklasse einschalten. Wie Sie wahrscheinlich wissen, passiert es allerdings selten, daß eine Init()-Methode überschrieben wird. Gewöhnlich erweitern Sie diese Methode mit zusätzlichem Code in einer abgeleiteten Klasse. Tatsächlich wird in einigen OOP-Sprachen der Code, der der Init()-Methode von VFP entspricht (und den die OOP-Programmierer die Constructor-Methode nennen), für jede Instanz der Klasse ausgeführt, sogar wenn Sie Prozeduren folgen lassen, die bewirken, daß dieser Code in anderen Methoden überschrieben wird.

Wo ist nun die Verbindung zum AutoCenter? Nehmen wir einmal an, Sie schalten AutoCenter in Ihrer Formularklasse ein. VFP schnappt sich die Größe des Formulars und errechnet die Position, in der das Formular im Hauptfenster von VFP fixiert werden soll. Und nun nehmen wir einmal an, daß Sie von dem Formular eine Klasse ableiten und dabei die Größe des Formulars mit Hilfe zusätzlichen Codes verändern – vielleicht sogar auf Faktoren basierend, die erst während der Laufzeit auftreten, so daß sie in der ursprünglichen nicht enthalten sind.

Wie Abbildung 2 zeigt, achtet ein Formular, bei dem AutoCenter eingeschaltet ist, nicht darauf, ob sich eine Größe geändert hat. Daher zentriert es sich auch nicht selbständig neu. (In der Abbildung habe ich die Eigenschaft _SCREEN.Borderstyle angepaßt, um zu betonen, daß ich die Größe meines Hauptfensters nicht geändert habe, um das Formular wieder zu zentrieren.) Wenn Sie in der Init()-Methode der abgeleiteten Klasse die Größe des Formulars festlegen, könnte ein nicht zentriertes Formular dabei herauskommen, wenn Sie nicht den Code erneut editieren, um AutoCenter auf .T. zu setzen.

Da das Erstellen von Formularklassen impliziert, daß Sie von ihnen jederzeit 

Abbildung 2

weitere Klassen ableiten können, schließe ich mich der allgemeinen Meinung an und schlage vor, daß Sie AutoCenter im Init() dieser Klassen nicht einschalten. Warten Sie lieber, bis Sie im Code die Klasse instantiieren, unabhängig davon, ob Sie die Syntax CREATEOBJECT(<Formularklasse>) oder DO FORM <SCXName> verwenden. Sie sollten AutoCenter nur benutzen, wenn die Klasse instantiiert ist – niemals im Code der Klasse.

Sag niemals nie

Ich danke meinem Kollegen und Mentor Jim Korenthal, der mich kürzlich daran erinnerte, daß ich häufig den Menschen rate, in Verbindung mit FoxPro niemals „nie“ zu sagen, um dann meinen eigenen Ratschlag zu vernachlässigen. Ich weiß, daß ich mit der Regel, die ich Ihnen gerade vorstellte, diesen Ratschlag anscheinend wieder ignoriert habe. Wenn Sie meine Herangehensweise an AutoCenter nicht mögen oder sie nicht durchgängig einsetzen können, dann stellen Sie wenigstens sicher, daß Sie AutoCenter nur in der letzten Instanz Ihrer Klassenhierarchie einsetzen. Setzen Sie diese Eigenschaft aber nicht in Ihrer Init()-Methode ein, da Sie sonst AutoCenter in jeder Instanz Ihrer Klassenhierarchie erneut setzen müssen, damit es arbeitet, da VFP die Position des Formulars in jeder Instanz erneut berechnen muß.

Seien Sie auch vorsichtig, wenn Sie AutoCenter oder andere Techniken der Positionierung des Formulars in anderen Methoden benutzen wie zum Beispiel Activate() oder Show(). Viele dieser Methoden werden ausgeführt, wenn das Formular sichtbar ist. Wenn Sie sich nicht vorsehen, hüpft das Formular plötzlich über den Bildschirm, wenn der Code ausgeführt wird. 

Manchmal müssen Sie ein Formular aber auch bewußt neu zentrieren, zum Beispiel wenn Sie die Größe des Formulars verändern oder der Anwender die Größe des Hauptfensters von VFP verändert. Das Formular stellt sein AutoCenter nicht automatisch auf .F., wenn eine solche Änderung durch das Programm oder den Anwender erfolgt, obwohl das Formular nicht mehr in der Mitte des Hauptfensters steht. Vielleicht ist der Name ja mißverständlich, aber die AutoCenter-Eigenschaft ermöglicht es dem Formular nicht, als Antwort auf Änderungen im Umfeld sich automatisch neu zu zentrieren, und sie schaltet sich auch nicht selbständig aus, um anzuzeigen, daß das Formular nicht länger im Mittelpunkt des Bildschirms steht.

Sie sollten nun wissen, daß AutoCenter immer dann ausgewertet wird, wenn Sie den Wert dieser Eigenschaft auf .T. setzen, auch wenn sie diesen Wert bereits besessen hat. Ihr Code kann jederzeit prüfen, ob die Größe des Formulars oder des Hauptfenster von VFP geändert wurde. Wenn eine solche Änderung stattgefunden hat, setzen Sie explizit die AutoCenter-Eigenschaft des Formulars auf .T., um das Formular erneut im Zentrum des Bildschirms zu positionieren.

Überlappende Formulare

Wie Sie sich erinnern versprach ich, daß die Techniken, die ich hier bespreche, aufeinander aufbauen. Passen Sie auf, was ich mit AutoCenter in M_FORM.PRG mache, dem Gegenstand dieses Abschnitts. Ich schrieb M_FORM.PRG, da die Optionen des Systemmenüs von VFP eine Unterlassung oder Unzulänglichkeit aufweisen. Der Eintrag im System-Menü _MWI_ARRAN („Alle anordnen“ im Menü Fenster) ordnet alle Fenster nebeneinander an. VFP stellt keine Menüoption zur Verfügung, die die Fenster überlappend anordnet.

Die Tatsache, daß das Nebeneinanderstellen der Fenster in VFP eingebaut ist, das Überlappen der Fenster aber nicht, ist recht ärgerlich. Die Fenster werden in ihrer Größe verändert und nebeneinandergestellt. Dies ist kein akzeptabler Weg, um Dokumentenfenster wie zum Beispiel Formulare für die Dateneingabe zu arrangieren. 

Abbildung 3

Abbildung 3 zeigt Ihnen fünf Instanzen eines einfachen Dateneingabeformulars, das auf der LABELS.DBF basiert, die Sie im HOME()-Verzeichnis von VFP finden. Vier Instanzen sind nebeneinandergestellt, so wie _MWI_ARRAN sie anpaßt und positioniert, um den Bildschirm auszunutzen. Eine fünfte Instanz (die über der Instanz unten links zu sehen ist) zeigt Ihnen das Formular, wie ich es entworfen habe, mit den drei Feldern in der Mitte des Fensters. Beachten Sie, daß in den von VFP angeordneten Fenstern die Zentrierung der Eingabefelder bei der Größenanpassung verlorengegangen ist.

Abbildung 4 zeigt Ihnen meine Alternative: überlappende Formulare. Beachten Sie, daß bei dem hier benutzten Mechanismus beim Arrangieren der Formulare die Proportionen erhalten bleiben. Hätte nun jedes dieser Formulare einen anderen Titel, der vielleicht anzeigt, welcher Datensatz in dem entsprechenden Formular editiert wird, könnte ein Anwender sich einfach durch diese Eingabesession bewegen.

Sie können Ihrem Menü eine Option für überlappende Fenster hinzufügen, indem Sie im Ergebnisfeld eines eigenen Menüs DO M_FORM.PRG eingeben. Wahlweise können Sie auch einen Code, der dem von M_FORM.PRG entspricht, in die Methode eines Objekts schreiben.

Listing A zeigt das gesamte M_FORM.PRG. Da M_FORM.PRG sich in einer Schleife durch _SCREEN.Forms() arbeitet, findet es 

Abbildung 4

sowohl Referenzen auf DEFINEd 

WINDOWs als auch auf Formular-Objekte, wie ich es früher schon einmal erläutert habe. Diese Fenster werden überlappend angeordnet, ohne daß zusätzlicher Code benötigt wird oder Probleme zu erwarten sind. 

* M_FORM.PRG

* Hilfsprogramm des MasterClass Framework,

* um existierende Formulare anzuordnen.

* Geändert für den Artikel Formulare - Tips und Tricks

* von Lisa Slater Nicholls

* Rückgabewert: Anzahl der Angeordneten Formulare

#DEFINE WINDOW_STAGGER_FACTOR   SYSMETRIC(9)

*  Höhe des Titels

LPARAMETERS tcFormName,tlOmitAutoCenteredForms,;





tnStartTop, tnStartLeft, tnStartColumn 

   LOCAL lnArranged, lnColumn, lnTop, lnLeft, loFormRef, ;

         lnParentHeight, lnScaleMode, lnIndex, llAllForms

   lnScaleMode = _SCREEN.ScaleMode

   _SCREEN.ScaleMode = 3

   lnArranged = 0

   lnTop =    IIF(TYPE("tnStartTop")  = "N", tnStartTop, 0)

   lnLeft =   IIF(TYPE("tnStartLeft") = "N", tnStartLeft,0)

   lnColumn = IIF(TYPE("tnStartColumn") = "N" AND tnStartColumn > 0, ;

                  tnStartColumn, 1)

   lnParentHeight = _SCREEN.Height

   llAllForms = EMPTY(tcFormName)

  FOR lnIndex = _SCREEN.FormCount TO 1 STEP -1

       loFormRef = _SCREEN.Forms(lnIndex)

       IF UPPER(loFormRef.BaseClass) == "TOOLBAR"

          LOOP

       ENDIF

       IF (llAllForms OR UPPER(loFormRef.Name) == UPPER(tcFormName)) AND ;

          loFormRef.WindowState = 0 AND loFormRef.Visible AND ;

          (NOT (tlOmitAutoCenteredForms AND loFormRef.AutoCenter))

          lnArranged = lnArranged + 1

          loFormRef.Top =  lnTop

          loFormRef.Left = (lnLeft * lnColumn)

          loFormRef.AutoCenter = .F.

          lnTop = lnTop + WINDOW_STAGGER_FACTOR

          lnLeft = lnLeft + WINDOW_STAGGER_FACTOR

          IF lnTop > lnParentHeight - WINDOW_STAGGER_FACTOR

             STORE WINDOW_STAGGER_FACTOR TO lnTop, lnLeft

             lnColumn = lnColumn + 1

          ENDIF

       ELSE

          * nichts zu tun

       ENDIF

   ENDFOR   

   _SCREEN.ScaleMode = lnScaleMode

RETURN lnArranged   

Listing M_FORM.PRG

Obwohl der Code als Voreinstellung alle Fenster, die in _SCREEN.Forms() referenziert werden, überlappend anordnet, können Sie M_FORM.PRG einen Parameter übergeben, mit dem Sie die Anordnung auf Fenster mit einem bestimmten Namen einschränken. Beispielsweise können Sie alle Instanzen einer bestimmten Klasse für die Erfassung von Daten jedes Mal automatisch anordnen, wenn Sie eine neue Instanz dieser Klasse starten, während die anderen Fenster so stehenbleiben, wie der Anwender sie sich arrangiert hat.

M_FORM.PRG empfängt noch einen zweiten optionalen Parameter, der es Ihnen erlaubt, zu entscheiden, wie mit Formularen verfahren werden soll, deren AutoCentered auf .T. gesetzt ist. Ohne diesen Parameter werden diese Formulare zusammen mit den anderen angeordnet. Wenn Sie es bevorzugen, können Sie aber auch M_FORM mitteilen, daß es alle Formulare mit gesetztem AutoCenter ignorieren soll, da Sie vermuten, daß diese bereits durch das Design an einer bestimmten Stelle stehen. Sie können Ihre Version von M_FORM.PRG auch so anpassen, daß alle Formulare mit eingeschaltetem AutoCenter neu zentriert werden, indem Sie die Eigenschaft neu initialisieren, wie ich es bereits früher ausgeführt habe.

Sie könnten jetzt den Wunsch haben, noch ein anderes Feature, das ich in M_FORM.PRG eingebaut habe, zu ändern: den Rand bzw. den Teil jedes Formulars, das der Anwender sieht, wenn die Formulare überlappend angeordnet sind. Ich habe mich entschieden, die Höhe des Titels, die von SYSMETRIC(9) zurückgegeben wird, als Wert zu nehmen, um wieviel die Formulare jeweils versetzt angezeigt werden sollen. Dieser Rand ermöglicht es dem Anwender, die Titel der Formulare leicht zu lesen und die Formulare durch Klicken auf den Titel auszuwählen. Sie können den Rand verändern, indem Sie den #DEFINEd WINDOW_STAGGER_FACTOR am Anfang des Programms ändern. Alternativ dazu können Sie auch für die horizontale und vertikale  Verschiebung des Formulars unterschiedliche Faktoren definieren

Positionieren eines WAIT WINDOW

Weiter oben habe ich Ihnen erzählt, daß Sie für ein WAIT WINDOW entsprechende Tricks für die Positionierung anwenden können. In VFP, wie in anderen Versionen von FoxPro, wird die Größe der WAIT WINDOWs dynamisch dem Inhalt angepaßt. Dadurch ist der Inhalt zwar immer gut lesbar, aber wir können die Größe nicht ändern, ohne an der Länge des Strings, der angezeigt werden soll, herumzupfuschen. VFP bietet uns eine neue AT-Klausel für das WAIT WINDOW-Kommando, die es uns ermöglicht, das WAIT WINDOW an jeder gewünschten Stelle zu plazieren, statt die Positionierung oben rechts in VFPs Hauptfenster akzeptieren zu müssen.

Abbildung 5

Leider ist die AT-Klausel des WAIT WINDOW so voller Überraschungen, daß Sie dieses neue Feature oft gar nicht einsetzen können, wenn Sie sie nicht mit zusätzlichem Code „einwickeln“. Wie Sie in Abbildung 5 sehen, wird das WAIT WINDOW ohne die AT-Klausel immer an einer bestimmten Position im Verhältnis zu VFPs Hauptfenster angezeigt.

 Abbildung 6 zeigt Ihnen, daß ein WAIT WINDOW mit der AT-Klausel relativ zum vollen Bildschirm positioniert wird, unabhängig von VFPs Hauptfenster.

Abbildung 6

Zusätzlich werden die Koordinaten der AT-Klausel relativ zu den Charakteristika von _SCREEN.Font ausgewertet. In Abbildung 6 wird das WAIT WINDOW von zwei Instanzen von VFP aufgerufen, die die gleichen Koordinaten (2,2) benutzen. Trotzdem befinden sich die beiden WAIT WINDOWs an unterschiedlichen Positionen, da ich in jeder Instanz von VFP unterschiedliche Werte für den FontSize benutzt habe. 

Es erscheint schon etwas merkwürdig, daß die Font-Charakteristika von _SCREEN Koordinaten bestimmen, die sich gar nicht im Bereich von _SCREEN befinden. Dieses Verhalten entspricht dem von FoxPro 2.6 für den Macintosh, wenn Sie SET MACDESKTOP auf ON setzen. Die AT-Klausel für ein WAIT WINDOW-Kommando beschreibt die Zeilen und Spalten (nicht die Pixel), genau wie die AT-Klausel eines DEFINE WINDOW- oder ACTIVATE POPUP-Kommandos. Zeilen und Spalten machen nur dann einen Sinn, wenn ihre Koordinaten relativ zu einem bestimmten Font und seinen Maßen (zum Beispiel Höhe und durchschnittliche Zeichenbreite) ausgedrückt sind. Immer wenn FoxPro Koordinaten relativ zum ganzen Bildschirm ausdrückt, nimmt es die Werte der aktuellen Eigenschaften seines eigenen Hauptfensters.

Zusätzlich hat ein WAIT WINDOW weder eine AutoCenter- noch eine ähnliche Eigenschaft. Es gibt kein spezielles Schlüsselwort, das Sie in der AT-Klausel benutzen können, und Sie können auch nicht das MOVE WINDOW CENTER-Kommando benutzen, da MOVE WINDOW nur mit DEFINEd WINDOWs zusammenarbeitet. Diese Unterlassung ist in Ordnung, da ein zentriertes WAIT WINDOW die einzige Verwendung der AT-Klausel ist, die nicht durch die Eigenschaften der Schrift oder des Bildschirms geändert wird. Tatsächlich gibt diese Einschränkung uns eine alternative konsistente Position für ein WAIT WINDOW, die wir zusätzlich zu der Standardposition bekommen, die wir auch ohne die AT-Klausel bereits hatten.

M_CWAIT.PRG gibt Ihnen einen Beispielcode, um Ihnen zu zeigen, wie Sie Schriftattribute anpassen und ein WAIT WINDOW zentrieren können. Wenn Sie WAIT WINDOW-Nachrichten benutzen, die sich über mehrere Zeilen erstrecken, und diese Zeilen mit Carriage Return trennen, werden Sie feststellen, daß mein Code die erste Zeile der Nachricht benutzt, um zu entscheiden, welche Ausdehnung das WAIT WINDOW braucht, um zentriert zu werden. Sie werden auch bemerken, daß die Zentrierung durch M_CWIND nicht perfekt ist, besonders, wenn die Ausdehnung einer Zeile stark von der durchschnittlichen Breite der Schrift abweicht.

* M_CWAIT.PRG

* Hilfsprogramm des MasterClass Framework um Wait Windows zu zentrieren

* Geändert für den Artikel Formulare - Tips und Tricks

* von Lisa Slater Nicholls

#DEFINE WAIT_WINDOW_FONT_NAME           "MS SANS SERIF"

#DEFINE WAIT_WINDOW_FONT_SIZE            8

#DEFINE WAIT_WINDOW_FONT_STYLE          "B"

#DEFINE WAIT_WINDOW_STANDARD_MESSAGE    "Press any key to continue ..."

#DEFINE WAIT_WINDOW_BORDER_WIDTH        SYSMETRIC(12)* 2   

LPARAMETERS tcMessage, tcAddtlParams, tcParentWindow 

* Der 2. Parameter sollte ein String wie "TIMEOUT 1" sein

IF TYPE("tcMessage") # "C"

   tcMessage = WAIT_WINDOW_STANDARD_MESSAGE

ENDIF

IF TYPE("tcAddtlParams") # "C"

   tcAddtlParams = ""

ENDIF

LOCAL lcFont, lnFont, lnScaleMode

lnFont = _SCREEN.FontSize

lcFont = _SCREEN.FontName

llFont = _SCREEN.FontBold

lnScaleMode = 3

_SCREEN.FontName = WAIT_WINDOW_FONT_NAME

_SCREEN.FontSize = WAIT_WINDOW_FONT_SIZE

_SCREEN.FontBold = ("B" $ WAIT_WINDOW_FONT_STYLE)

DO CASE

CASE TYPE("tcParentWindow") # "C"

  * Im Desktop zentriert; Vorgabewert

   WAIT WINDOW ;

        tcMessage ;

        &tcAddtlParams AT ;

        .5*( (SYSMETRIC(2) / ;

        (FONTM(1,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE) + ;

         FONTM(4,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE) + ;

         FONTM(5,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE))) - ;

         MEMLINES(tcMessage) ) ; 

         , ;

         .5*( ((SYSMETRIC(1)-WAIT_WINDOW_BORDER_WIDTH) / ;

         FONTM(6,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE)) - ;

         IIF(MEMLINES(tcMessage) > 1, ;

                 LEN(MLINE(tcMessage,1,0)), ;

                 LEN(tcMessage)) )

CASE EMPTY(tcParentWindow)

  * Zentrieren im Hauptfenster von FoxPro 

  * Berechnen des FoxPro-Desktop und

  * anschließendes Zentrieren

  LOCAL lnScaleMode

  lnScaleMode = _SCREEN.ScaleMode

  _SCREEN.ScaleMode = 0

   WAIT WINDOW ;

        tcMessage ;

        &tcAddtlParams AT ;

        .5*( _SCREEN.Height - MEMLINES(tcMessage) )  ;

        + _SCREEN.Top  + ;

        ( (SYSMETRIC(20) + SYSMETRIC(9)) / ;

        (FONTM(1,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE) + ;

         FONTM(4,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE) + ;

         FONTM(5,WAIT_WINDOW_FONT_NAME, ;

                 WAIT_WINDOW_FONT_SIZE, ;

                 WAIT_WINDOW_FONT_STYLE)))   ;

         , ;

         .5*( _SCREEN.Width   - ;

         IIF(MEMLINES(tcMessage) > 1, ;

                 LEN(MLINE(tcMessage,1,0)), ;

                 LEN(tcMessage)) ) + _SCREEN.Left

  _SCREEN.ScaleMode = lnScaleMode

CASE WEXIST(tcParentWindow)

  WAIT WINDOW NOWAIT ;

             tcMessage+ CHR(13)+CHR(13)+;

             "WAIT WINDOW centered in parent window "+ CHR(13)+ ;

             "is left as an exercise for the graduate student."

OTHERWISE

  * Fehlermeldung

ENDCASE

_SCREEN.FontName = lcFont

_SCREEN.FontSize = lnFont

_SCREEN.FontBold = llFont

RETURN

Listing M_CWAIT.PRG

M_CWAIT.PRG, das Sie in Listing B sehen, arbeitet mit den Merkwürdigkeiten von Schriftattributen. Der Code speichert als erstes die Charakteristika des Fonts des _SCREEN und ersetzt diese Attribute durch bekannte Werte, indem das Programm den Font und den Stil des WAIT WINDOW benutzt. Bevor die Funktion beendet wird, stellt sie die Originalwerte des _SCREEN wieder her.

Der erste Parameter von M_CWAIT enthält den Nachrichtentext für WAIT WINDOW. Wenn Sie diesen Parameter nicht übergeben, benutzt M_CWAIT die Defaultmessage von VFP. Sie sollten noch einen zweiten Parameter benutzen, um einen String mit zusätzlichen Parametern zu übergeben, zum Beispiel „TIMEOUT 1 TO memvar“, die Sie dem WAIT WINDOW-Kommando hinzufügen wollen.

In einem dritten optionalen Parameter können sie M_CWAIT einen Null-String übergeben, um das WAIT WINDOW im Hauptfenster von VFP zu zentrieren, genau wie WAIT WINDOW ohne den AT-Parameter positioniert wird. Wenn Sie diesen Parameter nicht übergeben, zentriert M_CWAIT das WAIT WINDOW im gesamten Desk Top und entspricht so Microsofts Implementierung der AT-Klausel für das WAIT WINDOW-Kommando. Abbildung 7 zeigt zwei Instanzen von VFP mit WAIT WINDOWS, die auf beide Arten zentriert worden sind. 

Abbildung 7

Hören Sie jetzt nicht auf

M_CWAIT.PRG führt Sie in die Komplexität der Schrift- und Systemattribute ein. Sie haben jetzt eine Vorstellung der relevanten Faktoren bei der Größenanpassung eines Formulars und seiner Positionierung. Wenn Sie sich diese Faktoren merken, sind Sie in der Lage, zusätzliche Techniken in allen Situationen einzusetzen, die das Verschieben und die Änderung der Größe eines Formulars erfordern.

